;; -*- coding: utf-8 -*-
(eval-when (:compile-toplevel :load-toplevel :execute)
  (sb-int:defconstant-eqx OPT
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0))
    #'equal)
  #+swank (ql:quickload '(:cl-debug-print :fiveam) :silent t)
  #-swank (set-dispatch-macro-character
           ;; enclose the form with VALUES to avoid being captured by LOOP macro
           #\# #\> (lambda (s c p) (declare (ignore c p)) `(values ,(read s nil nil t)))))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)
#-swank (disable-debugger) ; for CS Academy

;; BEGIN_INSERTED_CONTENTS
;; Should we do this with UNWIND-PROTECT?
(defmacro with-buffered-stdout (&body body)
  "Buffers all outputs to *STANDARD-OUTPUT* in BODY and flushes them to
*STANDARD-OUTPUT* after BODY has been done (without error). Note that only
BASE-CHAR is allowed."
  (let ((out (gensym)))
    `(let ((,out (make-string-output-stream :element-type 'base-char)))
       (let ((*standard-output* ,out))
         ,@body)
       (write-string (get-output-stream-string ,out)))))

(declaim (ftype (function * (values fixnum &optional)) read-fixnum))
(defun read-fixnum (&optional (in *standard-input*))
  (macrolet ((%read-byte ()
               `(the (unsigned-byte 8)
                     #+swank (char-code (read-char in nil #\Nul))
                     #-swank (sb-impl::ansi-stream-read-byte in nil #.(char-code #\Nul) nil))))
    (let* ((minus nil)
           (result (loop (let ((byte (%read-byte)))
                           (cond ((<= 48 byte 57)
                                  (return (- byte 48)))
                                 ((zerop byte) ; #\Nul
                                  (error "Read EOF or #\Nul."))
                                 ((= byte #.(char-code #\-))
                                  (setf minus t)))))))
      (declare ((integer 0 #.most-positive-fixnum) result))
      (loop
        (let* ((byte (%read-byte)))
          (if (<= 48 byte 57)
              (setq result (+ (- byte 48)
                              (* 10 (the (integer 0 #.(floor most-positive-fixnum 10)) result))))
              (return (if minus (- result) result))))))))

;;;
;;; Permutation and Combination
;;;

(declaim (inline map-permutations!))
(defun map-permutations! (function vector &optional (start 0) end)
  "Destructively permutes VECTOR[START] ... VECTOR[END-1] and applies FUNCTION
to VECTOR each time."
  (declare (vector vector))
  (labels ((recur (start end)
             (declare ((mod #.array-total-size-limit) start end))
             (if (> start end)
                 (funcall function vector)
                 (progn
                   (recur (1+ start) end)
                   (loop for i from (1+ start) below end
                         do (rotatef (aref vector start) (aref vector i))
                            (recur (1+ start) end)
                            (rotatef (aref vector start) (aref vector i)))))))
    (recur start (or end (length vector)))))

;; NOTE: It tends to be slow on SBCL version earlier than 1.5.0, as
;; constant-folding of ARRAY-ELEMENT-TYPE doesn't work.
(declaim (inline map-combinations))
(defun map-combinations (function vector length)
  "Applies FUNCTION to each combination of given length of VECTOR. Note that the
vector passed to FUNCTION will be recycled."
  (declare (vector vector)
           ((integer 0 #.most-positive-fixnum) length))
  (assert (<= length (length vector)))
  (let* ((n (length vector))
         (result (make-array length :element-type (array-element-type vector))))
    (declare ((mod #.array-total-size-limit) length))
    (labels ((recur (pos prev)
               (if (= pos length)
                   (funcall function result)
                   (loop for i from (1+ prev) below n
                         do (setf (aref result pos) (aref vector i))
                            (recur (1+ pos) i)))))
      (recur 0 -1))))

(declaim (inline map-permutations))
(defun map-permutations (function vector &optional k)
  "Applies FUNCTION to each permutation of length K of VECTOR. Note that the
vector passed to FUNCTION will be recycled."
  (declare (vector vector))
  (map-combinations (lambda (c) (map-permutations! function c)) vector k))

(defmacro do-permutations! ((var vector &optional (start 0) end) &body body)
  `(block nil (map-permutations! (lambda (,var) ,@body) ,vector ,start ,end)))

(defmacro do-combinations ((var vector length) &body body)
  `(block nil (map-combinations (lambda (,var) ,@body) ,vector ,length)))

(defmacro do-permutations ((var vector &optional k) &body body)
  `(block nil (map-permutations (lambda (,var) ,@body) ,vector ,k)))

(declaim (inline calc-lis))
(defun calc-lis (vector order infinity &optional restore)
  "Computes the lengths of LIS. Returns a LIS as a second value when RESTORE is
true."
  (declare (vector vector))
  (let* ((end 0)
         (dp (make-array (+ 1 (length vector))
                         :element-type (array-element-type vector)
                         :initial-element infinity)))
    (declare ((integer 0 #.array-total-size-limit) end))
    (labels ((bisect (ng ok value)
               (declare ((integer -1 #.array-total-size-limit) ng ok))
               (if (<= (- ok ng) 1)
                   ok
                   (let ((mid (ash (+ ng ok) -1)))
                     (if (funcall order value (aref dp mid))
                         (bisect ng mid value)
                         (bisect mid ok value))))))
      (sb-int:dovector (x vector)
        (let ((pos (bisect -1 end x)))
          (setf (aref dp pos) x)
          (when (= end pos)
            (incf end))))
      (if restore
          (let ((res (make-array end :element-type (array-element-type vector)))
                (pos (- end 1))
                (index (- (length vector) 1)))
            (declare ((integer -1 #.most-positive-fixnum)))
            (loop
              (let ((x (aref vector index)))
                (when (and (not (funcall order x (aref dp pos)))
                           (or (= pos (- end 1))
                               (funcall order x (aref res (+ pos 1)))))
                  (setf (aref res pos) x)
                  (decf pos))
                (when (< pos 0)
                  (return (values end res)))
                (decf index))))
          (values end nil)))))

(defmacro dbg (&rest forms)
  #+swank
  (if (= (length forms) 1)
      `(format *error-output* "~A => ~A~%" ',(car forms) ,(car forms))
      `(format *error-output* "~A => ~A~%" ',forms `(,,@forms)))
  #-swank (declare (ignore forms)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(declaim (inline println))
(defun println (obj &optional (stream *standard-output*))
  (let ((*read-default-float-format* 'double-float))
    (prog1 (princ obj stream) (terpri stream))))

(defconstant +mod+ 1000000007)

;;;
;;; Body
;;;

(defun test (n)
  (do-permutations! (perm (coerce (loop for i below n collect i) '(simple-array int32 (*))))
    (let ((lis-len (calc-lis perm #'< 100000000))
          (lds-len (calc-lis perm #'> -100000000)))
      (format t "~A ~D ~D~%" perm lis-len lds-len))))

(defun main ()
  (handler-bind ((error (lambda (c)
                          (declare (ignore c))
                          (println -1)
                          (return-from main))))
    (let* ((n (read))
           (a (read))
           (b (read))
           (dp (make-array a :element-type 'uint32 :initial-element 1)))
      (declare (uint31 n a b))
      (decf n a)
      (decf b)
      (dbg n a b)
      (incf (aref dp 0) b)
      (decf n b)
      #>dp
      (loop for i from 1 below a
            do (incf (aref dp i) (min n b))
               (decf n (min n b)))
      (assert (zerop n))
      (let ((num 1)
            (init t))
        (with-buffered-stdout
          (sb-int:dovector (delta dp)
            (loop for x from (+ num delta -1) downto num
                  do (if init
                         (setq init nil)
                         (write-char #\ ))
                     (princ x))
            (incf num delta)))))))

#-swank (main)

;;;
;;; Test and benchmark
;;;

#+swank
(defun io-equal (in-string out-string &key (function #'main) (test #'equal))
  "Passes IN-STRING to *STANDARD-INPUT*, executes FUNCTION, and returns true if
the string output to *STANDARD-OUTPUT* is equal to OUT-STRING."
  (labels ((ensure-last-lf (s)
             (if (eql (uiop:last-char s) #\Linefeed)
                 s
                 (uiop:strcat s uiop:+lf+))))
    (funcall test
             (ensure-last-lf out-string)
             (with-output-to-string (out)
               (let ((*standard-output* out))
                 (with-input-from-string (*standard-input* (ensure-last-lf in-string))
                   (funcall function)))))))

#+swank
(defun get-clipbrd ()
  (with-output-to-string (out)
    (run-program "C:/Windows/System32/WindowsPowerShell/v1.0/powershell.exe" '("get-clipboard") :output out)))

#+swank (defparameter *this-pathname* (uiop:current-lisp-file-pathname))
#+swank (defparameter *dat-pathname* (uiop:merge-pathnames* "test.dat" *this-pathname*))

#+swank
(defun run (&optional thing (out *standard-output*))
  "THING := null | string | symbol | pathname

null: run #'MAIN using the text on clipboard as input.
string: run #'MAIN using the string as input.
symbol: alias of FIVEAM:RUN!.
pathname: run #'MAIN using the text file as input."
  (let ((*standard-output* out))
    (etypecase thing
      (null
       (with-input-from-string (*standard-input* (delete #\Return (get-clipbrd)))
         (main)))
      (string
       (with-input-from-string (*standard-input* (delete #\Return thing))
         (main)))
      (symbol (5am:run! thing))
      (pathname
       (with-open-file (*standard-input* thing)
         (main))))))

#+swank
(defun gen-dat ()
  (uiop:with-output-file (out *dat-pathname* :if-exists :supersede)
    (format out "")))

#+swank
(defun bench (&optional (out (make-broadcast-stream)))
  (time (run *dat-pathname* out)))

;; To run: (5am:run! :sample)
#+swank
(it.bese.fiveam:test :sample
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "5 3 2
"
    "2 4 1 5 3
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "7 7 1
"
    "1 2 3 4 5 6 7
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "300000 300000 300000
"
    "-1
")))


Measured memory_bytes: 41060